
Toolbook sample

Parliament database

We will create a database of members of parliament.  The database will be searched and sorted and we will use some simple data management techniques such as calculations.  We will use date fields and investigate how to complete arithmetic with this data type.

The book will start with a picture of parliament.  Clicking on a seat will show a picture and electoral details of the member who occupies that seat.  Buttons on the first page will also allow the user to colour the seats of each party, search for a member and to produce a list of all members.

We will consider only 16 members.  The book has been started with a floor plan of parliament.  There are no buttons or script associated with the background of this first page.  Your first task is to load the book and add a new page with its own, new background.

Setting up the book

Load the file parlmnt.tbk from the network  U:\somewhere

Turn on the grid (0.20) and choose snap to

From the Object menu choose New Background
Go to the properties of this new background and name it   member_db

Place two rectangles on the background as shown below

Add Previous, Next and Parliament buttons

Write script for the previous and next buttons.  Name the buttons previous and next.

to handle buttonclick


send previous

end buttonclick

Write script for the Parliament button.  This button will return the user to the floor plan from the database records.

to handle buttonclick


go page “FloorPlan”

end buttonclick

You will need to use the recordfield tool 
  to create the database fields.  

Use the label tool 
  to provide a caption for the field.  

Inside the right hand rectangle, add the following ten labels and recordfields.  For each recordfield, set its Record Field Type property to Single Line.

Label
Recordfield name

Number:
seat_number

Position:
position

Surname:
surname

First name:
f_name

Political party:
party

Member for:
seat

First elected in:
first

Years in parliament:
years

Date of last rort:
rort_date

Years since last rort:
rort_years

Add grouped radio buttons for male and female.

Use the Text menu, Paragraph to select right alignment for all the labels.

Place a label to the right of the rort_date field and change the text to dd/mm/yy to indicate how this data is to be entered.



Select the “years” field and call up its properties.  Click on Activated to stop the user from entering data in this field.  Repeat for the rort_years field.

Go to the foreground and create 16 new pages (all using the member_db background).

Go to each of the 16 pages in turn and type the numbers 1 to 16 appropriately in the seat_number recordfield.  Name each page member1, member2, member 3…..member16.

Go to the background, select the seat_number field, click on the properties tool and choose Activated.  This stops the user from changing the number.



Adding remarks

Remarks can be added to the script by typing a double hyphen ‘--‘ then the comment. 

Initialise each page

Each page of the database needs to be have the recordfields blanked on entry if there is not a record already present.  We will accept that if there is something in the surname field then the record is still active.  The purpose of this step is to stop a partial deletion from impacting on future records.

The script will also disable the previous button on the first page and the next button on the last page.  It will make sure that both buttons are available on all other pages.  If the last bracket was not included then a button would be disabled if either the first or last page had recently been entered.

to handle enterPage


if this page is first page of this background then



enabled of button "previous" of this background = false



enabled of button "next" of this background = true


else



if this page is last page of this background then




enabled of button "previous" of this background = true




enabled of button "next" of this background = false



else




enabled of button "previous" of this background = true




enabled of button "next" of this background = true



end if


end if


if text of recordfield "surname" = "" then



checked of button "male" of this background = false



checked of button "female" of this background = false



text of recordfield "f_name" = ""



text of recordfield "party" = ""



text of recordfield "seat" = ""



text of recordfield "first" = 0



text of recordfield "years" = ""



text of recordfield "rort_date" = 0



text of recordfield "rort_years" = ""


end if


set focus to button "next" of this background

end enterPage

On the last page you will need to change the last line to read


set focus to button "previous" of this background

Link floor plan to member details

The first page the user will see will be the floor plan.  Clicking on a seat on the floor plan will open the record details of the member associated with the seat.



Each ‘seat’ will need to be named.  Name the field buttons 1, 2 …16 with the appropriate number.  For example, name the field button corresponding to the first seat ‘1’.

Enter the following script for each field button 1 to 16 of the floor plan.

to handle buttonclick


go page member1


-- this script for seat number 1

end buttonclick




-- change to member2… as appropriate

Allow for database records to be altered

The data in the nine fields of the database (other than the seat_number which is activated) can be changed by the user.  The following script makes sure that if the surname has been deleted that all remaining elements of the record are also deleted.  This is to safeguard against remnants of a previously deleted record impacting on a new record.

Enter the following script in the recordfield for surname on the background.

to handle leaveRecordField


if text of self = "" then



checked of button "male" of this background = false



checked of button "female" of this background = false



text of recordfield "position" = ""



text of recordfield "surname" = ""



text of recordfield "f_name" = ""



text of recordfield "party" = ""



text of recordfield "seat" = ""



text of recordfield "first" = 0



text of recordfield "years" = ""



text of recordfield "rort_date" = 0



text of recordfield "rort_years" = ""


end if

end leaveRecordField

A little arithmetic

The years in parliament assumes the “first” recordfield  contains the year in numeric data form.  The system date is used to determine today’s date.  The format of this date is changed to the year component only.  This is recognised as a numeric data form and can be subtracted from the “first” year.  Note the first line establishes a local variable ‘this_year’ which has a date data structure.  

Toolbook will not allow calculations to be performed on blank fields.  The self = “” code is to ensure that the field always has at least a zero in it.

Data validation is provided by an if() statement.

Enter the following script into the first recordfield:

to handle leaveRecordField


local date this_year


if text of self = "" then



text of self = 0


end if


sysDateFormat = "y"


elect = text of recordfield "first"


if elect < 1950 then



request "error message"


else



this_year = sysDate



years_in = this_year - elect



text of recordfield "years" = years_in


end if


end leaveRecordField

The calculation of the years since last rort requires a comparison of two dates.  Toolbook requires date arithmetic to be performed in ‘seconds’.  The sysDateFormat command is used to set all date fields in the same format.  The d/m/yy format recognises dates as 23/6/98 while d/m/y would return 23/6/1998.

There are 86400 seconds in a day.  There are likely to be more efficient ways of completing this code than indicated here.

Enter the following script into the rort_date recordfield:

to handle leaveRecordField


if text of self = "" then



text of self = 0


end if


local date today


local date rort


local date no_years


sysDateFormat = "d/m/yy"


rort = text of recordfield "rort_date" 


today = sysDate


sysDateFormat = "seconds"


no_years = today - rort


num_yrs = no_years/86400/365.25


format num_yrs as "0.00"


text of recordfield "rort_years" = num_yrs

end leaveRecordField

Both the first and the rort_date recordfields have a default value of zero set when a new page is entered or a current record has been deleted.  This is cumbersome when you wish to enter data into these fields since you must first delete the zero before data entry.  The following code should be entered into both fields to facilitate data entry:

to handle enterRecordField


text of self = ""

end enterRecordField

Keyboard events

To facilitate data entry the enter key can be used to progress to the next recordfield.  The keycode for the enter key is 13.

In the surname recordfield script enter the following code under the leaveRecordField script.

to handle keychar keypressed


if keypressed = 13 then



if text of self = "" then




if pagenumber of this page = last page of this background then





set focus to button "previous" of this background




else





set focus to button "next" of this background




end if




forward



else




set focus to recordfield "f_name"



end if


else



forward


end if

end keychar

The forward statement is critical.  Without this the program would not allow any further keyboard activity.  It passes the keychar message up to the next object in the Toolbook hierarchy.  The handler here is overriding the default keyboard behaviour.

Copy and alter appropriately this script into the party, seat and first recordfields.

Copy the code with slight differences to the f_name and rort_date recordfields.

For f_name replace the ‘set focus of recordfield’ line with 


set focus to button "male" of this background

For rort_date replace the ‘set focus of recordfield’ line with


set focus to button "next" of this background

Enter data

Enter appropriate data in several/most of the pages.  You will need to be in ‘reader’ level.  The libs are on the left, labour on the right and the greens you can put anywhere.  You will need some data before the remaining sections can work.  Make sure you are consistent with party names.

Add a picture into the left rectangle on each member page.

Colour the seats corresponding to each party

Go to page one, the floor plan.

Create three buttons in the top left corner of the page and label them ‘Liberal’, ‘Labour’ and ‘Greens’.  Change the colour of each button to blue, red and green respectively.  Name the buttons ‘show_libs’, ‘show_labour’ and ‘show_greens’.



We will write code that will colour the seat associated with each party.  Enter the following code in the script for the ‘liberal’ button:

to handle buttonclick


step counter from 1 to 16



if text of recordfield "party" of page ("member"&counter) = "liberal" then




fillcolor of field counter = blue


-- color not colour



end if


end step

end buttonclick

Note the use of ‘&’ to join the expression ‘member’ with the current counter value to derive the page name.

Repeat for the labour  (red) and greens (green) party.

We need to clear the colours each time we return to the floor plan.  Otherwise, when a member switches parties the colour from the last execution of this code will prevail.  Make sure you are looking at the floor plan in author mode with no fields selected.  Enter the following code:

to handle enterPage


step counter from 1 to 16





fillcolor of field counter = white


end step

end enterPage

We will also ensure that all buttons are white when the application is entered.

From the Object menu select Book Properties.  Select Script and enter the following:

to handle enterApplication


step counter from 1 to 16





fillcolor of field counter of page "FloorPlan" = white


end step


sysLevel = reader

end enterApplication

The last line ensures that the application is always in ‘reader’ mode when booted up.

Searching the database

We will create a button that will search the database for the surname of a member.

Make sure you are in author mode and looking at the floor plan.

Create a button in the right corner of the page.  Name and caption the button ‘Find member’.  Colour the button.



Enter the following code:

to handle buttonclick


ask "Which member do you wish?"


step counter from 1 to 16





if text of recordfield "surname" of page ("member"&counter) = It then




go page ("member"&counter)




query = "found"



else




if text of recordfield "seat" of page ("member"&counter) = It then





go page ("member"&counter)





query = "found"




end if



end if


end step


if query <> "found" then



request "Member not present"


end if

end buttonclick

The ‘ask’ command will produce a dialog box containing the message which follows.  If the OK button on the resulting dialog box is pressed the string entered into the dialog box by the user is assigned by the system to a variable ‘It’.



The script includes a local variable ‘query’ which is used to register whether a search string is present.  The script will search the surname and the seat recordfields for the text string entered by the user.  The request command produces a dialog box containing a message for the user.

Test that the button is working for both surnames and seats.  Add the code necessary for the script to also check for position.  This will allow the user to search for details of members if they know the surname or the seat held or the ministry held by the member.



List of records

Create a new button in the right corner and name and caption it ‘Member list’.  Colour the button.



Create a viewer and name it ‘list_viewer’.  (Object menu Viewers …. New …Read-only Popup … Create New Page and Background)

Go to the viewer properties and set the size to 10cm by 10 cm and the position to centred.

Go to the page associated with the viewer (probably page 18).  Name this page ‘page_of_viewer’. Go to the background of this page and in its properties set the page size to ‘set to viewer’.

Return to the foreground and to ‘page_of_viewer’.  Add five text fields.

In the first field set up a title “List of members”.  Set the font, size and alignment appropriately.

In the next field enter ‘Member’, ‘Seat’ and ‘Party’ appropriately.

Name the three adjacent fields ‘member_list’, ‘seat_list’ and ‘party_list’.  Set the font to TNR 10 pt and the alignment to left in all three fields.

Return to the floor plan page and select the ‘Member list button’

Enter the following script:

to handle buttonclick


local namelast[]


-- sets up an array


local namefirst[]


-- for each of these


local member_seat[]

-- variables


local member_party[]


newline = ansitochar(10)

-- ansitochar converts ansi number to keypress char


found = 0


text of field "member_list" of page "page_of_viewer" = ""
-- initialises fields to blank


text of field "seat_list" of page "page_of_viewer" = ""
-- so no remnants appear


text of field "party_list" of page "page_of_viewer" = ""


step counter from 1 to 16



if text of recordfield "surname" of page ("member"&counter) <> "" then




found = found +1




namelast[found] = text of recordfield "surname" of page ("member"&counter)




namefirst[found] = text of recordfield "f_name" of page ("member"&counter)




member_seat[found] = text of recordfield "seat" of page ("member"&counter)




member_party[found] = text of recordfield "party" of page ("member"&counter)



end if


end step


step counter from 1 to found



member_details= namelast[counter]&", "&namefirst[counter]



seat_details = member_seat[counter]



party_details = member_party[counter]



text of field "member_list" of page "page_of_viewer" = text of field\



 "member_list" of page "page_of_viewer" & member_details & newline



text of field "seat_list" of page "page_of_viewer" = text of field\



 "seat_list" of page "page_of_viewer" & seat_details & newline



text of field "party_list" of page "page_of_viewer" = text of field\



 "party_list" of page "page_of_viewer" & party_details & newline


end step


text of field "member_list" of page "page_of_viewer" = text of field\



 "member_list" of page "page_of_viewer" & newline



text of field "seat_list" of page "page_of_viewer" = text of field\



 "seat_list" of page "page_of_viewer" & newline


text of field "party_list" of page "page_of_viewer" = text of field\



 "party_list" of page "page_of_viewer" & newline


show viewer "list_viewer"



-- result is displayed in a popup viewer

end buttonclick





-- it could have been placed in a page without viewer

Test the book

You should check that all components of the program are working.

( Colin Potts – Trinity Grammar

